A Simple Weir Structure Floodflow Bypass Analysis Program

William V. Burchard (1) and T.V. Hromadka II (2)

(1) Engineer, Williamson and Schmid, Irvine, CA, USA
(2) Director of Water Resources, Williamson and Schmid, Irvine, CA, USA

ABSTRACT

A simple interactive weir structure floodflow bypass analysis program is developed to estimate the amount of water that will spill over a weir wall for a given inflow into the channel. This provides the user the ability to estimate the optimum weir wall height to meet the design requirements, by calculating the channel outflow and weir capture flows for a given channel inflow. Use of this program aids in developing a cost-to-benefit curve for subsequent optimization. The purpose of this paper is to provide information on the use and practical application of the weir overflow analysis program. A detailed example of how this program works is provided in the example problem by showing how the program was implemented into the design of the Galivan Retarding Basin.

INTRODUCTION

Use of the weir overflow analysis program provides the ability to create a cost-to-benefit curve. The advantage of developing a cost-to-benefit curve is that the optimum design in terms of efficiency and cost is determined. In the case of the Galivan Retarding Basin, the cost of excavation of the basin was compared with the cost of construction of the weir wall. The cost-to-benefit curve indicates the optimum weir length and excavation of the retarding basin.

In the following example problem, a method for developing a cost-to-benefit curve is described. The example problem's design purpose is to transfer excess flows from Oso Creek into the Galivan Retarding Basin via a weir structure. The design conditions are to restrict the Q_{100} flows of 8700 cfs to 4200 cfs, transferring 4500 cfs into the retarding basin (see Fig. 1). Since Oso Creek is to be fully developed as a rectangular channel, it was decided to place a weir wall within the channel with an optimum height and length.

EXAMPLE PROBLEM:

OPTIMIZING BASIN EXCAVATION VS. WEIR STRUCTURE COSTS

Use of an optimum weir length will result in less excess flows being transferred into the basin, resulting in less excavation. Since Oso Creek parallels the retarding basin and continues to extend upstream from the basin in a straight line, the opportunity to design an optimum weir length is available (see Fig. 2).

The length of the weir was determined by comparing the cost of excavation and any savings realized in excavation by the reduction in excess flows with the cost of any given weir length. This was accomplished by first calculating the cost of construction of various weir lengths and heights. To achieve this, the following equation was used:

$$c\left(\frac{hwl}{27}\right) = C$$



INFLOW VS. OUTFLOW HYDROGRAPH FOR ALTERNATIVE 6 WEIR DESIGN

Fig. 1. Inflow vs. Outflow for Galivan Retarding Basin

0266-9463/86/020215 - 04 £2.00



Fig. 2. Plan view of weir structure and detention of basin system

where

- c = cost of concrete and rebar per cubic yard (approx. $250)
- h = height of weir wall (varies with length)
- w = width of weir wall (1.5 foot)
- l = length of weir wall
- C = cost of weir wall

After selecting several arbitrary weir lengths their corresponding optimum weir heights were estimated by varying the weir height in the hydraulic calculations for each arbitrary weir length chosen. The weir height selected was the one which allowed flows closest to but not exceeding 4200 cfs to pass by the weir.

The results were plotted on the Cost Optimization figure (Fig. 3) and are represented by the solid straight line. It should be noted that as the length of the weir wall increased, the height of the weir wall also increased. With the increased height, only flows closer to the peak flow would be able to travel over the weir. With the increased length less excess flows were captured.

Second, the cost of excavation based upon weir length was calculated. This was accomplished by choosing several weir lengths and calculating the volume of water that would flow over in each case. This volume of water indicated the amount of storage necessary. From this information the amount of excavation per cubic yard of dirt could be calculated. Once calculated, the result was plotted on the Cost Optimization figure and are represented by the solid curved line.

Finally, the costs of both weir construction and excavation per weir length were added together. The results were plotted on the Cost Optimization figure and are represented by the dashed curved line. The lowest point on this curve represents the optimum costs vs. weir length and indicates the optimum weir length to be approximately 850 LF. With the optimum weir length chosen, the optimum weir height could be estimated. It was shown that, with S_o = 0.0040 and a weir coefficient of 3.0, the optimum weir height for the 850-foot weir is 5.05.



Fig. 3. Cost Optimization Figure

WEIR PROGRAM DESCRIPTION

The weir program solves for the continuity equation along segments of the weir by balancing the inflow of water to the segment to the outflow of water from the segment due to flow over the segment's weir structure.

The weir structure geometry is assumed to be composed of simple straightline transitions (see Fig. 4). The weir structure is of a constant height and the continuing channel is of constant slope, so the channel width varies linearly from a beginning width to a terminating width. Finally, flows over the weir are assumed given by the weir relationship of

$$Q_i = C\, W_i d^{1.5} \quad (1)$$

where

- Q_i = weir flow over segment i
- C = constant weir coefficient 0<C<3.09
- W_i = width of segment i
- d = flowdepth of water above weir at segment i

The program initiates by checking the minimal flow through the channel system assuming normal depth. If the entered flowrate can pass through the weir structure without overflow, the program returns to the main menu of commands.

If the entered flowrate does encounter weir overflow, the overflow is estimated using the weir flow relationship of (1) and Fig. 4. Based on the entered segment width length, the weir structure is subdivided into segments for sequential analysis. On a segment-by-segment basis, the weir overflow is estimated by use of (1) and the flowdepth corresponding to normal depth in the rectangular channel cross-section based on the channel slope, flowrate, Manning's friction factor, and channel width. After estimating Q_i, this flowrate is subtracted from the segment's upstream flowrate to produce the next segment's upstream flowrate. Proceeding in this fashion, the analysis proceeds downstream until the flowrate is either less than the capacity of the channel (without weir overflows), or the end of the structure is reached. The example problem illustrates the calculations.



Fig. 4. Weir Structure Geometry

The program returns the user to the main menu after each analysis in order to provide an interactive mode.

PROGRAM-READ-WRITE UNITS

The program is set up for the following unit numbers in the format statements:

TABLE 1. WEIR PROGRAM UNIT DESCRIPTIONS

Unit Number	Description
10	Write to CRT
11	Read from CRT

EXAMPLE PROBLEM

A weir structure has the following dimensions and data:

weir structure length	100 feet
channel slope	0.0050 feet/feet
weir structure beginning width	30 feet
weir structure end width	20 feet
channel Manning's friction factor	0.015
weir height	4.0 feet
weir coefficient	2.5

After accepting the weir structure data entries, a flow value is entered for subsequent analysis. For this example, a flow of 500 cfs and a segment spacing of 10 feet results in a maximum normal depth of 2.332 feet which is less than the 4-foot weir height; hence, no weir overflow is estimated.

For 2,000 cfs (and a segment spacing of 10 feet), however, the following results are developed:

TABLE 2. EXAMPLE PROBLEM RESULTS

Distance From Start (feet) -1-	Flowrate (cfs) -2-	Channel Width (feet) -3-	Excess Height (feet) -4-	Spillover (cfs) -5-
0.0	2000.0	30.0	0.26	3.3
10.0	1996.7	29.0	0.38	5.9
20.0	1990.8	28.0	0.48	8.3
30.0	1982.5	27.0	0.60	11.7
40.0	1970.8	26.0	0.72	15.4
50.0	1955.4	25.0	0.85	19.5
60.0	1935.9	24.0	0.97	23.8
70.0	1912.1	23.0	1.12	29.4
80.0	1882.7	22.0	1.24	34.4
90.0	1848.3	21.0	1.38	40.7

In the above tabulation, it is seen that the weir program calculations are based on the segments upstream normal depth hydraulics. Consequently, for the first segment, normal depth is calculated at column -1- using the flowrate of column -2- and channel width of column -3-. For the first segment, normal depth at the upstream end is 4.26 feet; hence, 0.26 foot of flowdepth is above the weir. For C=2.5, equation (1) gives a spillover of 3.3 cfs (column -5-). The 2000 cfs flowrate is then reduced to 1996.7 cfs (column -2-).

By using different segment widths, calculation effort varies, and so does relative accuracy. Table 3 compares the output results for several segment sizes.

TABLE 3. WEIR PROGRAM RESULTS VS. SEGMENT WIDTHS (Example Problem)

Segment Width (ft)	Number of Segments	Final Outflow (cfs)
100	1	1966.8
50	2	1877.6
25	4	1831.8
10	10	1807.6
5	20	1798.4
1	100	1792.0
0.10	1000	1789.7

APPENDIX: WEIR OVERFLOW PROGRAM

```
C        PROGRAM TO ESTIMATE CONTINUITY EQUATION FOR WEIR OVERFLOW
C
C
100      CONTINUE
         WRITE(10,1001)
1001     FORMAT(10X,"THIS PROGRAM ESTIMATES THE CONTINUITY",/,
      C  10X,"EQUATION FOR FLOW OVER A WEIR",////)
         WRITE(10,1002)
1002     FORMAT(10X,"ENTER WEIR STRUCTURE LENGTH(FEET):",/)
         READ FREE(11)XL
         WRITE(10,1003)
1003     FORMAT(////,10X,"ENTER WEIR STRUCTURE CHANNEL SLOPE(FT/FT):",/)
         READ FREE(11)S0
         WRITE(10,1004)
1004     FORMAT(////,10X,"ENTER WEIR STRUCTURE CHANNEL BEGINNING",
      C  " WIDTH(FT):",/)
         READ FREE(11)W1
         WRITE(10,1005)
1005     FORMAT(////,10X,"ENTER WEIR STRUCTURE CHANNEL END WIDTH(FT):",/)
         READ FREE(11)W2
         WRITE(10,1006)
1006     FORMAT(////,10X,"ENTER WEIR STRUCTURE CHANNEL MANNINGS ",
      C  "FACTOR:",/)
         READ FREE(11)XN
         WRITE(10,1007)
1007     FORMAT(////,10X,"ENTER WEIR HEIGTH(FT):")
         READ FREE(11)H
         WRITE(10,10071)
10071    FORMAT(///,10X,"ENTER WEIR COEFFICIENT:",/)
         READ FREE(11)XC
         WRITE(10,1008)XL,S0,W1,W2,XN,H,XC
1008     FORMAT(///////,5X,"ENTERED DATA:",/,
      C  10X,"WEIR LENGTH(FT) = ",F10.3,/,
      C  10X,"CHANNEL SLOPE(FT/FT) = ",F10.5,/,
      C  10X,"BEGINNING CHANNEL WIDTH(FT) = ",F10.3,/,
      C  10X,"END CHANNEL WIDTH(FT) = ",F10.3,/,
      C  10X,"CHANNEL MANNINGS FACTOR = ",F10.5,/,
      C  10X,"WEIR HEIGTH(FT) = ",F10.3,/,
      C  10X,"WEIR COEFFICIENT =",F10.4,///,
      C  10X,"ENTER DATA OPTION NUMBER:",/,
      C  5X,"1 = ACCEPT DATA AND CONTINUE",/,
      C  5X,"2 = REJECT DATA AND REENTER DATA VALUES",/,
      C  5X,"3 = EXIT PROGRAM",/)
         READ FREE(11)KOPT
         IF(KOPT.EQ.3)GOTO 10000
         IF(KOPT.EQ.2)GOTO 100
200      CONTINUE
         WRITE(10,1050)
1050     FORMAT(/////////,10X,"ENTER PROGRAM OPTION NUMBER;",/,
      C  5X,"1 = ENTER FLOW VALUE INTO STRUCTURE",/,
      C  5X,"2 = REDEFINE WEIR STRUCTURE SYSTEM",/,
      C  5X,"3 = EXIT PROGRAM",/)
         READ FREE(11)KOPT
         IF(KOPT.EQ.2)GOTO 100
         IF(KOPT.EQ.3)GOTO 10000
         WRITE(10,1010)
1010     FORMAT(////,10X,"ENTER WEIR STRUCTURE INFLOW(CFS):",/)
         READ FREE(11)Q
         WRITE(10,1011)

1011     FORMAT(///,10X,"ENTER CALCULATION SPACING ALONG WEIR(FT):",/)
         READ FREE(11)DD
         CALL THORMA(S0,Q,0.,W2,XN,V,10,DN,100.)
         IF(DN.GT.H)GOTO 210
         WRITE(10,1012)DN,H
1012     FORMAT(//,5X,"*** FLOWDEPTH OF ",F10.3," IS LESS THAN",/,
      C  5X,"WEIR HEIGTH OF ",F10.3,".... FLOW PASSES THROUGH STRUCTURE.")
         GOTO 200
C...FLOW SPILLS OVER WEIR......
210      DW=(W2-W1)/XL
         DW=DW*DD
C....INITIALIZE DO-LOOP VARIABLES
         WT=W1-DW
         DQ=0.
         QQ=Q
         XSUM=-DD
         WRITE(10,1019)
1019     FORMAT(11X,"DISTANCE  FLOWRATE   CHANNEL   EXCESS   SPILLOVER"
      C  ,/,10X,"FROM START   (CFS)      WIDTH    HEIGHT      (CFS)")
250      CONTINUE
         WT=WT+DW
         QQ=QQ-DQ
         XSUM=XSUM+DD
         IF(XSUM.LT.XL)GOTO 260
         GOTO 200
260      CALL THORMA(S0,QQ,0.,WT,XN,V,10,DN,100.)
         DH=DN-H
         DQ=0.
         IF(DH.LE.0.)GOTO 250
         DQ=XC*DD*DH**1.5
         WRITE(10,1020)XSUM,QQ,WT,DH,DQ
1020     FORMAT(10X,5F10.3)
         GOTO 250
10000    CONTINUE
         STOP
         END
```

```
         SUBROUTINE THORMA(S0,Q,Z,B,XN,V,NUT,DN,DMAX)
         DN=DMAX/2.
         YMIN=0.
         YMAX=DMAX
         FUN=1.-Q*XN*(B+2.*DMAX*SQRT(Z*Z+1.))**.6667/(1.486*((B+Z*DMAX)
      C  *DMAX)**1.6667*SQRT(S0))
         IF(FUN)450,500,500
450      WRITE(NUT,460)
460      FORMAT(/,10X,'==>>ERROR: FLOW IN CHANNEL EXCEEDS CHANNEL',/,
      C  14X,'CAPACITY( NORMAL DEPTH EQUAL TO SPECIFIED MAXIMUM ',/,
      C  14X,'ALLOWABLE DEPTH).',
      C  /,14X,'AS AN APPROXIMATION, FLOWDEPTH IS SET AT MAXIMUM',/,
      C  14X,'ALLOWABLE DEPTH AND IS USED FOR TRAVELTIME CALCULATIONS.',/)
         DN=DMAX
         GOTO 570
500      I=0
510      I=I+1
         IF(I.GT.30)GO TO 560
         FUN=1.-Q*XN*(B+2.*DN*SQRT(Z*Z+1))**.6667/(1.486*((B+Z*DN)
      C*DN)**1.6667*SQRT(S0))
         IF(FUN)520,540,530
520      YMIN=DN
         GO TO 540
530      YMAX=DN
540      DN1=(YMIN+YMAX)/2.
         IF(ABS(DN-DN1).LT..01)GO TO 570
         DN=DN1
550      GO TO 510
560      WRITE(NUT,2010)
2010     FORMAT(/,10X,'==>>ERROR: NO CONVERGENCE IN PROGRAM TO NORMAL DEPTH.',
      C  /,14X,'AS AN APPROXIMATION, FLOWDEPTH IS SET AT MAXIMUM ALLOWABLE',/,
      C  14X,'DEPTH AND IS USED FOR TRAVELTIME CALCULATIONS.',/)
         DN=DMAX
570      CONTINUE
         TA=(DN*Z+B)*DN
         V=Q/TA
3000     CONTINUE
         RETURN
         END
```